

Rank v Excelu představuje úlohu zjišťování pořadí hodnoty v nějakém výčtu (seznamu), který je v našem případě uveden na listu. Pro dané účely nabízí Excel funkci RANK, novodobě pak RANK.EQ (Excel 2010+) a RANG. AVG (Excel 2010+). A rovnou předesílám, že v případě sestavování výsledkových listin sportovních událostí jsou nám k ničemu. Druhá zmíněná je jen novodobým přepisem první, ta poslední se hodí pouze v případě zjišťování tzv. statistického pořadí. Pojdme si udělat malou analýzu.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1														
2				vzestupně ... nižší hodnota = nižší (lepší) pořadí, viz časy ve sportu										
3														
4		Data									Usp. data	Pořadí		
5		50		1	5	5	5	5,5	5,5		10	1		D5: =COUNTIF(\$B\$5:\$B5;\$B5)
6		10		1	1	1	1	1	1		20	2		E5: =RANK(B5;\$B\$5:\$B\$14;1)
7		40		1	4	4	4	4	4		30	3		F5: =RANK.EQ(B5;\$B\$5:\$B\$14;1)
8		30		1	3	3	3	3	3		40	4		G5: =RANK(B5;\$B\$5:\$B\$14;1)+COUNTIF(\$B\$5:\$B5;\$B5)-1
9		20		1	2	2	2	2	2		50	5		H5: =RANK.AVG(B5;\$B\$5:\$B\$14;1)
10		50		2	5	5	6	5,5	5,5		50	5		I5: =(2*RANK(B5;\$B\$5:\$B\$14;1)+COUNTIF(\$B\$5:\$B\$14;\$B5)-1)/2
11		70		1	8	8	8	8	8		60	6		
12		90		1	10	10	10	10	10		70	7		K5: =SMALL(\$B\$5:\$B\$14;ŘÁDEK(A1))
13		80		1	9	9	9	9	9		80	8		L5: =SOUČIN.SKALÁRNÍ((K5>K\$5:K\$14)/COUNTIF(K\$5:K\$14;K\$5:K\$14))+1
14		60		1	7	7	7	7	7		90	9		
15														
16				sestupně ... vyšší hodnota = nižší (lepší) pořadí, viz body ve sportu										
17														
18		Data									Usp. data	Pořadí		
19		50		1	5	5	5	5,5	5,5		90	1		D19: =COUNTIF(\$B\$19:\$B19;\$B19)
20		10		1	1	1	1	1	1		80	2		E19: =RANK(B19;\$B\$19:\$B\$28;1)
21		40		1	4	4	4	4	4		70	3		F19: =RANK.EQ(B19;\$B\$19:\$B\$28;1)
22		30		1	3	3	3	3	3		60	4		G19: =RANK(B19;\$B\$19:\$B\$28;1)+COUNTIF(\$B\$19:\$B19;\$B19)-1
23		20		1	2	2	2	2	2		50	5		H19: =RANK.AVG(B19;\$B\$19:\$B\$28;1)
24		50		2	5	5	6	5,5	5,5		50	5		I19: =(2*RANK(B19;\$B\$19:\$B\$28;1)+COUNTIF(\$B\$19:\$B\$28;\$B19)-1)/2
25		70		1	8	8	8	8	8		40	6		
26		90		1	10	10	10	10	10		30	7		K19: =LARGE(\$B\$19:\$B\$28;ŘÁDEK(A1))
27		80		1	9	9	9	9	9		20	8		L19: =SOUČIN.SKALÁRNÍ((K19<K\$19:K\$28)/COUNTIF(K\$19:K\$28;K\$19:K\$28))+1
28		60		1	7	7	7	7	7		10	9		
29														

RANK v Excelu - analýza

Funkce COUNTIF nám vrátí počet výskytů dané hodnoty ve výčtu. Následují vestavěné funkce typu RANK. Vzorce ve sloupci I představují náhradu funkce RANK.AVG pro ranější verze Excelu. Sloupce K a L jsou pak určeny pro sportovní výsledky (pořadí setříděných hodnot).

Rank pro sportovní výsledky

V první řadě si musíme uvědomit, je-li naším cíle uspořádání hodnot vzestupně nebo sestupně. Ve sportu o tomto rozhoduje hodnocení podle dosaženým časů (čím kratší, tím lepší, přímá úměra) nebo dosažených bodů (čím více, tím lépe, nepřímá úměra). Nejde vám to na rozum? Slovo „lepší“ přeci v obou případech značí „nižší pořadí“. Úvaha se pak promítne do posledního parametru funkcí typu RANK (0 nebo 1). Jak je bohužel vidět, dané funkce neřeší duplicitní časy či body a tedy pořadí, kdy

závodníci se stejným časem či skóre (v příkladu hodnota 50) obsadí stejnou příčku (což je nepříjemné hlavně ve smyslu neznámého počtu vyrobených medailí), a další v pořadí se umístí na pozici o jedničku vyšší. K tomu si už musíme dopomoci funkcí COUNTIF (pozor na adresování v jejich parametrech!) a následně buď maticovým vzorcem typu SUMA(KDYŽ(...)) nebo nematicovou formou a funkcí SOUČIN.SKALÁRNÍ. Předložené řešení přitom aplikujeme na setříděná data (funkce SMALL, resp. LARGE). Je to proto, že v praxi budou zpravidla nesetříděná data (časy, body) uložena do startovní listiny, zatímco naše pořadí zapíšeme vedle do výsledkové listiny. Ale to není vše. Podívejte se na trochu učesanou formu a sportovní výsledky.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		Startovní listina				Výsledková listina				Výsledková listina			
3													
4		Pořadí	Závodník	Hodnota		Pořadí	Závodník	Body		Pořadí	Závodník	Čas	
5		1.	Jméno 01	50		1.	Jméno 08	90		1.	Jméno 02	10	
6		2.	Jméno 02	10		2.	Jméno 09	80		2.	Jméno 05	20	
7		3.	Jméno 03	40		3.	Jméno 07	70		3.	Jméno 04	30	
8		4.	Jméno 04	30		4.	Jméno 10	60		4.	Jméno 03	40	
9		5.	Jméno 05	20		5.	Jméno 01	50		5.	Jméno 01	50	
10		6.	Jméno 06	50		5.	Jméno 06	50		5.	Jméno 06	50	
11		7.	Jméno 07	70		6.	Jméno 03	40		6.	Jméno 10	60	
12		8.	Jméno 08	90		7.	Jméno 04	30		7.	Jméno 07	70	
13		9.	Jméno 09	80		8.	Jméno 05	20		8.	Jméno 09	80	
14		10.	Jméno 10	60		9.	Jméno 02	10		9.	Jméno 08	90	
15													
16		F5: =SOUČIN.SKALÁRNÍ((H5<H\$5:H\$14)/COUNTIF(H\$5:H\$14;H\$5:H\$14))+1											
17		G5: {=INDEX(\$C\$5:\$C\$14;SMALL(KDYŽ(\$D\$5:\$D\$14=H5;ŘÁDEK(\$D\$5:\$D\$14)-MIN(ŘÁDEK(\$D\$5:\$D\$14))+1);COUNTIF(\$H\$5:\$H5;\$H5)))}											
18		H5: =LARGE(\$D\$5:\$D\$14;\$B5)											
19													
20		J5: =SOUČIN.SKALÁRNÍ((L5>L\$5:L\$14)/COUNTIF(L\$5:L\$14;L\$5:L\$14))+1											
21		K5: {=INDEX(\$C\$5:\$C\$14;SMALL(KDYŽ(\$D\$5:\$D\$14=L5;ŘÁDEK(\$D\$5:\$D\$14)-MIN(ŘÁDEK(\$D\$5:\$D\$14))+1);COUNTIF(\$L\$5:\$L5;\$L5)))}											
22		L5: =SMALL(\$D\$5:\$D\$14;\$B5)											
23													

RANK v Excelu – sportovní výsledky

Ať už tak či tak, nejprve v tabulce řešíme body (časy), následně pořadí, a nakonec doplňující informace o závodníkovi. A zde znovu narážíme na problém v případě duplicitního skóre (časů). Pokud se chcete probrat maticovými vzorci ve sloupcích G a K, doporučuji prostudovat článek [Get nth match](#), kdy hledáme n-tý výskyt položky v seznamu. Další údaje k závodníkovi (startovní číslo, klub apod.) už je hračka pro funkci typu SVYHLEDAT. Tolik know-how k dané problematice. Ani to ovšem v praxi není konečná.

Při navrhování aplikace pro jeden sportovní klub jsem narazil na další problémy. Uvedené vzorce

nemají rády prázdné buňky. A kde by se tam vzali? Nevíte předem, kolik účastníků soutěž či závod absolvuje (odhlášení ze závodu nemusí být řešeno náhradníkem). Čili na startovní listině se může objevit bílé místo. Proto je výhodné na startovní listině aplikovat definované názvy (dynamicky se odkazující pouze na vyplněná data). Aby toho nebylo málo, v průběhu závodu může dojít k diskvalifikaci nebo nedokončení závodu účastníkem (v zápisu času či bodů se najednou objeví text typu DNF). Zkrátka se může stát, že nakonec se umlátíte při ošetřování výjimek s funkcí KDYŽ... Přidejte si k tomu tisk výsledkových listin s proměnným počtem řádků, funkčnost přidělování startovních čísel, vyhodnocování přes skupiny (kluby, země, věkové kategorie) a máte o zábavu vyláskáno :-)

Ke stažení

[rank_sport.zip](#)

Fráze pro Google

„Rank duplicate values with duplicate ranks without skipping numbers“